Pensieve header: This is the main Mathematica package that goes along with the paper “A Very Fast, Very Strong, Topologically Meaningful and Fun Knot Invariant” by Dror Bar-Natan and Roland van der Veen.

In[∘]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\Theta"];
```

tex

We start by loading the package \verb$KnotTheory`$ --- it is only needed because it has many specific knots pre-defined:

pdf

In[∘]:=
```
<< KnotTheory`
```

pdf

```
Loading KnotTheory` version of October 29, 2024, 10:29:52.1301.
Read more at http://katlas.org/wiki/KnotTheory.
```

tex

Next we quietly define the commands \verb$Rot$, used to compute rotation numbers, and \verb$PolyPlot$, used to plot polynomials as bar codes and as hexagonal QR codes. Neither is a part of the core of the computation of Θ, so neither is shown; yet we do show some usage examples.

pdf

In[∘]:=
```
(* Rot suppressed *)
```

In[∘]:=
```
Rot[pd_PD] := Module[{n, xs, x, rots, Xp, Xm, front = {1}, k},
    n = Length@pd; rots = Table[0, {2 n}];
    xs = Cases[pd, x_X :> { Xp[x⟦4⟧, x⟦1⟧]   PositiveQ@x
                            Xm[x⟦2⟧, x⟦1⟧]   True        ];
    For[k = 1, k ≤ 2 n, ++k,
     If[FreeQ[front, -k],
      front = Flatten@Replace[front, k → (xs /. {
              Xp[k, l_] | Xm[l_, k] :> {l + 1, k + 1, -l},
              Xp[l_, k] | Xm[k, l_] :> (++rots⟦l⟧; {-l, k + 1, l + 1}),
              _Xp | _Xm :> {}
            }), {1}],
      Cases[front, k | -k] /. {k, -k} :> --rots⟦k⟧;
     ]
    ];
    {xs /. {Xp[i_, j_] :> {+1, i, j}, Xm[i_, j_] :> {-1, i, j}}, rots} ];
Rot[K_] := Rot[PD[K]];
```

pdf

In[∘]:=
```
Rot[Mirror@Knot[3, 1]]
```

Out[∘]=

pdf

```
{{{1, 1, 4}, {1, 3, 6}, {1, 5, 2}}, {0, 0, 0, -1, 0, 0}}
```

tex

We urge the reader to compare the above output with the knot diagram in Section~\ref{ssec:OldFormulas}.

pdf

In[ ]:=
```
(* PolyPlot suppressed *)
```

```
In[∘]:= PolyPlot1[Δ_] := Module[{crs, m, maxc, minc, s, rect},
    rect = {{0, 0}, {1, 0}, {1, 1}, {0, 1}};
    If[Expand[Δ] === 0, Graphics[],
     m = Max[-Exponent[Δ, T, Min], Exponent[Δ, T, Max]];
     crs = CoefficientRules[T^m Δ, {T}];
     maxc = N@Log@Max@Abs[Last /@ crs];
     minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
     If[minc == maxc, s[_] = 0, s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
     Graphics[crs /. ({x_} → c_) :> {
         Lighter[Which[c == 0, White, c > 0, Red, c < 0, Blue], 0.88 s[Abs@c]],
         Tooltip[Polygon[({x + m - 1/2, 0} + #) & /@ rect], c T^(x-m)]
        }, AspectRatio → Min[1/5, 1/√(m + 1)],
      ImagePadding → None, PlotRangePadding → None]
    ]];
Options[PolyPlot2] = {Labeled → False};
PolyPlot2[θ_, OptionsPattern[]] := Module[{crs, m1, m2, maxc, minc, s, hex, p},
    If[Expand[θ] === 0, Graphics[{White, Disk[]}],
     hex = Table[{Cos[α], Sin[α]} / Cos[2 π / 12] / 2, {α, 2 π / 12, 2 π, 2 π / 6}];
     m1 = Max[-Exponent[θ, T_1, Min], Exponent[θ, T_1, Max]];
     m2 = Max[-Exponent[θ, T_2, Min], Exponent[θ, T_2, Max]];
     crs = CoefficientRules[T_1^m1 T_2^m2 θ, {T_1, T_2}];
     maxc = N@Log@Max@Abs[Last /@ crs];
     minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
     If[minc == maxc, s[_] = 0, s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
     Graphics[{(*{Yellow,Disk[{0,0},1+Cos[2π/12]Norm[{m1,m2}]/√2]},*)
       crs /. ({x1_, x2_} → c_) :> {
          Lighter[Which[c == 0, White, c > 0, Red, c < 0, Blue], 0.88 s[Abs@c]],
          p = ({{1, -1/2}, {0, √3/2}}).{x1 - m1, x2 - m2};
          Tooltip[Polygon[(p + #) & /@ hex], c T_1^(x1-m1) T_2^(x2-m2)] ,
          If[Not@OptionValue[Labeled], {}, {Black, Text[Style[c T_1^(x1-m1) T_2^(x2-m2), 30], p]}]
         }
      }, ImagePadding → None, PlotRangePadding → None]
    ]];
PolyPlot[{Δ_, θ_}, opts___Rule] := GraphicsColumn[
   {PolyPlot1[Δ], PolyPlot2[θ, FilterRules[{opts}, Options[PolyPlot2]]]},
   Spacings → Scaled@0.08, ImagePadding → None, PlotRangePadding → None,
   FilterRules[{opts}, Options[GraphicsColumn]]
  ];
```

In[∘]:= $\texttt{PPDemo = PolyPlot2}\left[2+T_1-T_1\,T_2+T_2-T_1^{-1}+T_1^{-1}\,T_2^{-1}-T_2^{-1},\ \texttt{Labeled}\to\texttt{True}\right]$

Out[∘]=



In[∘]:=
```
Export["PPDemo.pdf", PPDemo]
```

Out[∘]=

```
PPDemo.pdf
```

In[∘]:= $1/2+T_1-T_1\,T_2+T_2-T_1^{-1}+T_1^{-1}\,T_2^{-1}-T_2^{-1}$ // TeXForm

Out[∘]//TeXForm=

```
-T_2 T_1+T_1+T_2-\frac{1}{T_2}+\frac{1}{T_2 T_1}-\frac{1}{T_1}+\frac{1}{2}
```

pdf

In[∘]:= $\texttt{PolyPlot}\left[\left\{T-1+T^{-1},\ T_1+2\,T_2-4\,T_1^{-1}\,T_2^{-1}\right\},\ \texttt{ImageSize}\to\texttt{Tiny}\right]$

Out[∘]=
pdf



tex

We urge the reader to reflect on how the ``QR Code'' part of the above picture corresponds to the 2-variable polynomial $T_1+2T_2-4T_1^{-1}T_2^{-1}$.

Next, we decree that $T_3=T_1T_2$ and define the three ``Feynman Diagram'' polynomials F_1, F_2, and F_3 (those definitions are printed in a smaller font because they are equal to what was already printed in \eqref{eq:F1}--\eqref{eq:F3}):

pdf

In[◦]:=
```
T3 = T1 T2;

CF[ℰ_] := Module[{vs = Union@Cases[ℰ, g__, ∞], ps, c},
   Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) ⧴ Factor[c] (Times @@ vs^ps)]];
```

exec

```
nb2tex$PDFWidth *= 1.5
```

In[◦]:=
```
CF[s (1/2 - g3ii + T2^s g1ii g2ji - g1ii g2jj - (T2^s - 1) g2ji g3ii + 2 g2jj g3ii - (1 - T3^s) g2ji g3ji -
     g2ii g3jj - T2^s g2ji g3jj + g1ii g3jj + ((T1^s - 1) g1ji (T2^(2 s) g2ji - T2^s g2jj + T2^s g3jj) +
        (T3^s - 1) g3ji (1 - T2^s g1ii - (T1^s - 1) (T2^s + 1) g1ji + (T2^s - 2) g2jj + g2ij)) / (T2^s - 1)) /. s → 1]
```

Out[◦]=

$$\frac{1}{2} + T_2\, g_{1,i,i}\, g_{2,j,i} + \frac{(-1+T_1)\, T_2^2\, g_{1,j,i}\, g_{2,j,i}}{-1+T_2} - g_{1,i,i}\, g_{2,j,j} - \frac{(-1+T_1)\, T_2\, g_{1,j,i}\, g_{2,j,j}}{-1+T_2} - g_{3,i,i} + (1-T_2)\, g_{2,j,i}\, g_{3,i,i} + 2\, g_{2,j,j}\, g_{3,i,i} + \frac{(-1+T_1 T_2)\, g_{3,j,i}}{-1+T_2} - \frac{T_2\,(-1+T_1 T_2)\, g_{1,i,i}\, g_{3,j,i}}{-1+T_2} - \frac{(-1+T_1)\,(1+T_2)\,(-1+T_1 T_2)\, g_{1,j,i}\, g_{3,j,i}}{-1+T_2} + \frac{(-1+T_1 T_2)\, g_{2,i,j}\, g_{3,j,i}}{-1+T_2} + (-1+T_1 T_2)\, g_{2,j,i}\, g_{3,j,i} + \frac{(-2+T_2)\,(-1+T_1 T_2)\, g_{2,j,j}\, g_{3,j,i}}{-1+T_2} + g_{1,i,i}\, g_{3,j,j} + \frac{(-1+T_1)\, T_2\, g_{1,j,i}\, g_{3,j,j}}{-1+T_2} - g_{2,i,i}\, g_{3,j,j} - T_2\, g_{2,j,i}\, g_{3,j,j}$$

pdf

In[◦]:=

$$F_1[\{1, i\_, j\_\}] = \frac{1}{2} + T_2\, g_{1ii}\, g_{2ji} + \frac{(T_1-1)\, T_2^2\, g_{1ji}\, g_{2ji}}{T_2-1} - g_{1ii}\, g_{2jj} - \frac{(T_1-1)\, T_2\, g_{1ji}\, g_{2jj}}{T_2-1} - g_{3ii} + (1-T_2)\, g_{2ji}\, g_{3ii} + 2\, g_{2jj}\, g_{3ii} + \frac{(T_3-1)\, g_{3ji}}{T_2-1} - \frac{T_2\,(T_3-1)\, g_{1ii}\, g_{3ji}}{T_2-1} - \frac{(T_1-1)\,(T_2+1)\,(T_3-1)\, g_{1ji}\, g_{3ji}}{T_2-1} + \frac{(T_3-1)\, g_{2ij}\, g_{3ji}}{T_2-1} + (T_3-1)\, g_{2ji}\, g_{3ji} + \frac{(T_2-2)\,(T_3-1)\, g_{2jj}\, g_{3ji}}{T_2-1} + g_{1ii}\, g_{3jj} + \frac{(T_1-1)\, T_2\, g_{1ji}\, g_{3jj}}{T_2-1} - g_{2ii}\, g_{3jj} - T_2\, g_{2ji}\, g_{3jj};$$

In[◦]:=
```
CF[s (1/2 - g3ii + T2^s g1ii g2ji - g1ii g2jj - (T2^s - 1) g2ji g3ii + 2 g2jj g3ii - (1 - T3^s) g2ji g3ji -
     g2ii g3jj - T2^s g2ji g3jj + g1ii g3jj + ((T1^s - 1) g1ji (T2^(2 s) g2ji - T2^s g2jj + T2^s g3jj) +
        (T3^s - 1) g3ji (1 - T2^s g1ii - (T1^s - 1) (T2^s + 1) g1ji + (T2^s - 2) g2jj + g2ij)) / (T2^s - 1)) /. s → -1]
```

pdf

In[∘]:=
$$F_1[\{-1, i\_, j\_\}] = -\frac{1}{2} - \frac{g_{1ii}\, g_{2ji}}{T_2} - \frac{(T_1-1)\, g_{1ji}\, g_{2ji}}{T_1 (T_2-1) T_2} + g_{1ii}\, g_{2jj} + \frac{(T_1-1)\, g_{1ji}\, g_{2jj}}{T_1 (T_2-1)} + g_{3ii} - \frac{(T_2-1)\, g_{2ji}\, g_{3ii}}{T_2} - 2\, g_{2jj}\, g_{3ii} - \frac{(T_3-1)\, g_{3ji}}{T_1 (T_2-1)} + \frac{(T_3-1)\, g_{1ii}\, g_{3ji}}{T_1 (T_2-1) T_2} - \frac{(T_1-1)(T_2+1)(T_3-1)\, g_{1ji}\, g_{3ji}}{T_1^2 (T_2-1) T_2} - \frac{(T_3-1)\, g_{2ij}\, g_{3ji}}{T_1 (T_2-1)} + \frac{(T_3-1)\, g_{2ji}\, g_{3ji}}{T_3} + \frac{(2 T_2-1)(T_3-1)\, g_{2jj}\, g_{3ji}}{T_1 (T_2-1) T_2} - g_{1ii}\, g_{3jj} - \frac{(T_1-1)\, g_{1ji}\, g_{3jj}}{T_1 (T_2-1)} + g_{2ii}\, g_{3jj} + \frac{g_{2ji}\, g_{3jj}}{T_2};$$

In[∘]:=
$$\mathrm{CF}\left[s1 \left(T_1^{s0}-1\right)\left(T_2^{s1}-1\right)^{-1}\left(T_3^{s1}-1\right) g_{1,j1,i0}\, g_{3,j0,i1} \left(\left(T_2^{s0} g_{2,i1,i0} - g_{2,i1,j0}\right) - \left(T_2^{s0} g_{2,j1,i0} - g_{2,j1,j0}\right)\right) \,/.\, \{s0 \to 1, s1 \to 1\}\right]$$

pdf

In[∘]:=
$$F_2[\{1, i0\_, j0\_\}, \{1, i1\_, j1\_\}] = \frac{(T_1-1)\, T_2 (T_3-1)\, g_{1,j1,i0}\, g_{2,i1,i0}\, g_{3,j0,i1}}{T_2-1} - \frac{(T_1-1)(T_3-1)\, g_{1,j1,i0}\, g_{2,i1,j0}\, g_{3,j0,i1}}{T_2-1} - \frac{(T_1-1)\, T_2 (T_3-1)\, g_{1,j1,i0}\, g_{2,j1,i0}\, g_{3,j0,i1}}{T_2-1} + \frac{(T_1-1)(T_3-1)\, g_{1,j1,i0}\, g_{2,j1,j0}\, g_{3,j0,i1}}{T_2-1};$$

In[∘]:=
$$\mathrm{CF}\left[s1 \left(T_1^{s0}-1\right)\left(T_2^{s1}-1\right)^{-1}\left(T_3^{s1}-1\right) g_{1,j1,i0}\, g_{3,j0,i1} \left(\left(T_2^{s0} g_{2,i1,i0} - g_{2,i1,j0}\right) - \left(T_2^{s0} g_{2,j1,i0} - g_{2,j1,j0}\right)\right) \,/.\, \{s0 \to 1, s1 \to -1\}\right]$$

pdf

In[∘]:=
$$F_2[\{1, i0\_, j0\_\}, \{-1, i1\_, j1\_\}] = -\frac{(T_1-1)\, T_2 (T_3-1)\, g_{1,j1,i0}\, g_{2,i1,i0}\, g_{3,j0,i1}}{T_1 (T_2-1)} + \frac{(T_1-1)(T_3-1)\, g_{1,j1,i0}\, g_{2,i1,j0}\, g_{3,j0,i1}}{T_1 (T_2-1)} + \frac{(T_1-1)\, T_2 (T_3-1)\, g_{1,j1,i0}\, g_{2,j1,i0}\, g_{3,j0,i1}}{T_1 (T_2-1)} - \frac{(T_1-1)(T_3-1)\, g_{1,j1,i0}\, g_{2,j1,j0}\, g_{3,j0,i1}}{T_1 (T_2-1)};$$

In[∘]:=
$$\mathrm{CF}\left[s1 \left(T_1^{s0}-1\right)\left(T_2^{s1}-1\right)^{-1}\left(T_3^{s1}-1\right) g_{1,j1,i0}\, g_{3,j0,i1} \left(\left(T_2^{s0} g_{2,i1,i0} - g_{2,i1,j0}\right) - \left(T_2^{s0} g_{2,j1,i0} - g_{2,j1,j0}\right)\right) \,/.\, \{s0 \to -1, s1 \to 1\}\right]$$

pdf

In[∘]:=
$$F_2[\{-1, i0\_, j0\_\}, \{1, i1\_, j1\_\}] = -\frac{(T_1-1)(T_3-1)\, g_{1,j1,i0}\, g_{2,i1,i0}\, g_{3,j0,i1}}{T_1 (T_2-1) T_2} + \frac{(T_1-1)(T_3-1)\, g_{1,j1,i0}\, g_{2,i1,j0}\, g_{3,j0,i1}}{T_1 (T_2-1)} + \frac{(T_1-1)(T_3-1)\, g_{1,j1,i0}\, g_{2,j1,i0}\, g_{3,j0,i1}}{T_1 (T_2-1) T_2} - \frac{(T_1-1)(T_3-1)\, g_{1,j1,i0}\, g_{2,j1,j0}\, g_{3,j0,i1}}{T_1 (T_2-1)};$$

In[∘]:=
$$\mathrm{CF}\left[s1 \left(T_1^{s0}-1\right)\left(T_2^{s1}-1\right)^{-1}\left(T_3^{s1}-1\right) g_{1,j1,i0}\, g_{3,j0,i1} \left(\left(T_2^{s0} g_{2,i1,i0} - g_{2,i1,j0}\right) - \left(T_2^{s0} g_{2,j1,i0} - g_{2,j1,j0}\right)\right) \,/.\, \{s0 \to -1, s1 \to -1\}\right]$$

pdf

In[∘]:=
```
F2[{-1, i0_, j0_}, {-1, i1_, j1_}] =
```
$$\frac{(T_1-1)(T_3-1)\, g_{1,j1,i0}\, g_{2,i1,i0}\, g_{3,j0,i1}}{T_1^2 (T_2-1) T_2} - \frac{(T_1-1)(T_3-1)\, g_{1,j1,i0}\, g_{2,i1,j0}\, g_{3,j0,i1}}{T_1^2 (T_2-1)} - \frac{(T_1-1)(T_3-1)\, g_{1,j1,i0}\, g_{2,j1,i0}\, g_{3,j0,i1}}{T_1^2 (T_2-1) T_2} + \frac{(T_1-1)(T_3-1)\, g_{1,j1,i0}\, g_{2,j1,j0}\, g_{3,j0,i1}}{T_1^2 (T_2-1)};$$

pdf

In[∘]:=
```
F3[φ_, k_] = φ g3kk - φ / 2;
```

exec

```
nb2tex$PDFWidth /= 1.5
```

tex

Next comes the main program computing Θ. Fortunately, it matches perfectly with the mathematical description in Section~\ref{sec:Formulas}. In line 01 we let X be the list of crossings in an input knot K, and φ the list of its rotation numbers, using the external program \verb$Rot$ which we have already mentions. We also let n be the length of X, namely, the number of crossings in K. In line 02 we let the starting value of A be the identity matrix, and then in line 03, for each crossing in X we add to A a 2×2 block, in rows i and j and columns $i+1$ and $j+1$, as explain in Equation~\eqref{eq:A}. In line 04 we compute the normalized Alexander polynomial Δ as in~\eqref{eq:Delta}. In line 05 we let G be the inverse of A. In line 06 we declare what it means to evaluate, \verb$ev$, a formula $\calE$ that may contain symbols of the form $g_{\nu\alpha\beta}$: each such symbol is to be replaced by the entry in position α,β of G, but with T replaced with T_ν. In line 07 we start computing θ by computing the first summand in~\eqref{eq:Main}, which in itself, is a sum over the crossings of the knot. In line 08 we add to θ the double sum corresponding to the second term in~\eqref{eq:Main}, and in line 09, we add the third summand of~\eqref{eq:Main}. Finally, line 10 outputs a pair: Δ, and the renormalized version of θ.

pdf

In[∘]:=
```
Θ[K_] := Θ[K] = Module[{X, φ, n, A, Δ, G, ev, θ},
   (* 01 *) {X, φ} = Rot[K]; n = Length[X];
   (* 02 *) A = IdentityMatrix[2 n + 1];
   (* 03 *) Cases[X, {s_, i_, j_} :→ (A⟦{i, j}, {i + 1, j + 1}⟧ += ( -T^s  T^s - 1 ; 0  -1 ))];
   (* 04 *) Δ = T^((-Total[φ] - Total[X⟦All,1⟧])/2) Det[A];
   (* 05 *) G = Inverse[A];
   (* 06 *) ev[ℰ_] := Factor[ℰ /. g_{ν_,α_,β_} :→ (G⟦α, β⟧ /. T → T_ν)];
   (* 07 *) θ = ev[∑_{k=1}^{n} F₁[X⟦k⟧]];
   (* 08 *) θ += ev[∑_{k1=1}^{n} ∑_{k2=1}^{n} F₂[X⟦k1⟧, X⟦k2⟧]];
   (* 09 *) θ += ev[∑_{k=1}^{2 n} F₃[φ⟦k⟧, k]];
   (* 10 *) Factor@{Δ, (Δ /. T → T₁) (Δ /. T → T₂) (Δ /. T → T₃) θ}
  ];
```

tex

On to examples! Starting with the trefoil knot.

pdf

In[∘]:=
```
Expand[Θ[Knot[3, 1]]]
PolyPlot[Θ[Knot[3, 1]], ImageSize → Tiny]
```

pdf

In[∘]:=
```
Θ[Knot[3, 1]]
```

pdf

KnotTheory: Loading precomputed data in PD4Knots`.

Out[∘]=

pdf

$$\left\{\frac{1 - T + T^2}{T}, -\frac{1 - T_1 + T_1^2 - T_2 - T_1^3 T_2 + T_2^2 + T_1^4 T_2^2 - T_1 T_2^3 - T_1^4 T_2^3 + T_1^2 T_2^4 - T_1^3 T_2^4 + T_1^4 T_2^4}{T_1^2 T_2^2}\right\}$$



tex

\parpic[r]{\parbox{42mm}{
\includegraphics[width=20mm]{figs/K11n34.png}\hfill\includegraphics[width=20mm]{figs/K11n42.p-
ng}
}}

Next are the Conway knot 11_{n34} and the Kinoshita-Terasaka knot 11_{n42}. The two are mutants and famously hard to separate: they both have $\Delta=1$ (as evidenced by their one-bar bar codes below), and they have the same HOMFLY-PT polynomial and Khovanov homology. Yet their θ invariants are different. Note that the genus of the Conway knot is 3, while the genus of the Kinoshita-Terasaka knot is 2. This agrees with the apparent higher complexity of the QR code of the Conway polynomial, and with the observations in Section~\ref{sec:SandM}.

pdf

In[∘]:=
```
PolyPlot[Θ[Knot[#]], ImageSize → Small] & /@ {"K11n34", "K11n42"}
```

pdf

KnotTheory: Loading precomputed data in DTCode4KnotsTo11`.

pdf

KnotTheory: The GaussCode to PD conversion was written by Siddarth Sankaran at the University of Toronto in the summer of 2005.

Out[∘]=
pdf



tex

Torus knots have particularly nice-looking Θ invariants. Here are the torus knots $T_{13/2}$, $T_{17/3}$, $T_{13/5}$, and $T_{7/6}$:

pdf

In[ ]:=
```
GraphicsGrid[{
  TubePlot[TorusKnot @@ #] & /@ {{13, 2}, {17, 3}, {13, 5}, {7, 6}},
  PolyPlot[Θ[TorusKnot @@ #]] & /@ {{13, 2}, {17, 3}, {13, 5}, {7, 6}}
 }]
```

Out[ ]=
pdf



tex

The next line shows the computation time is seconds for the 132-crossing torus knot $T_{22/7}$ on a 2024 laptop, without actually showing the output. The output plot is in Figure~\ref{fig:T227}.

pdf

In[ ]:=
```
AbsoluteTiming[Θ[TorusKnot[22, 7]];]
```

Out[ ]=
pdf

{715.344, Null}

In[ ]:=
```
AbsoluteTiming[Θ[Mirror@TorusKnot[22, 7]];]
```

Out[ ]=

{1222.55, Null}

tex

\begin{figure}

pdf

In[•]:=
```
ImageCompose[PolyPlot[Θ[TorusKnot[22, 7]], ImageSize → 720],
  TubePlot[TorusKnot[22, 7], ImageSize → 360], {Right, Bottom}, {Right, Bottom}]
```

Out[•]=
pdf



tex

\caption{The 132-crossing torus knot $T_{22/7}$ and a plot of its Θ invariant} \label{fig:T227} \end{figure}

In[•]:=
```
tab250 = {{1, 0}}~Join~Table[Θ[K], {K, AllKnots[{3, 10}]}];
```

In[∘]:=
```
g250 = GraphicsGrid[Partition[PolyPlot /@ tab250, 10],
  Spacings → 0, Dividers → All, ImagePadding → None, PlotRangePadding → None]
```

Out[∘]=



In[∘]:=
```
Export["Theta4Rolfsen.pdf", g250]
```

Out[∘]=

```
Theta4Rolfsen.pdf
```

See also https://drorbn.net/AcademicPensieve/Projects/HigherRank/DunfieldKnots/.

In[∘]:=
```
DunfieldKnots = ReadList["../../People/Dunfield/nmd_random_knots"] /. k_Integer :→ k + 1;
DK[n_] := DunfieldKnots⟦n - 2⟧;
```

In[∘]:=
```
AbsoluteTiming[th = Θ[DK[50]];]
PolyPlot[th]
```

Out[∘]=
```
{21.4991, Null}
```

Out[∘]=



Proof of R3

exec
```
nb2tex$TeXFileName = "Invariance-R3.tex";
```

tex

First, we implement the Kronecker δ-function, the g-rules for a crossing (s,i,j), and the g-rules for a list of crossings X:

pdf

In[∘]:=
```
δ_{i_,j_} := If[i === j, 1, 0];
gRules[{s_Integer, i_, j_}] := {
    g_{ν_ jβ_} :→ g_{ν j⁺ β} + δ_{jβ}, g_{ν_ iβ_} :→ T_ν^s g_{ν i⁺ β} + (1 - T_ν^s) g_{ν j⁺ β} + δ_{iβ},
    g_{ν_ α_ i⁺} :→ T_ν^s g_{ναi} + δ_{αi⁺}, g_{ν_ α_ j⁺} :→ g_{ναj} + (1 - T_ν^s) g_{ναi} + δ_{αj⁺}
  };
gRules[{X___List}] := Union @@ Table[gRules[c], {c, {X}}]
```

tex

We then let \verb$Xl$ be the three crossings in the left-hand-side of the R3 move, as in Figure~\ref{-

fig:R3}, we let \verb$Al$ be the A^l term of~\eqref{eq:ABC}, and we let \verb$lhs$ be the result of applying the g-rules for the crossings in \verb$Xl$ to \verb$Al$. We print only a ``\verb$Short$'' version of \verb$lhs$ because the full thing would cover about 2.5 pages:

pdf

In[∘]:=
```
Xl = {{1, j, k}, {1, i, k⁺}, {1, i⁺, j⁺}};
Al = Sum[F₁[c], {c, Xl}] + Sum[F₂[c0, c1], {c0, Xl}, {c1, Xl}];
lhs = Simplify[Al //. gRules[Xl]];
Short[lhs, 5]
```

Out[∘]//Short=
pdf

$$-\frac{1}{2\,(1-T_2)}\,(3-3\,T_2+\ll 128\gg+2\,(1-T_2)\,T_2\,g_{2,(k^+)^+,i}\,(1+(1-T_1\,T_2)\,g_{3,(k^+)^+,j}+g_{3,(k^+)^+,k})+2\,(1-T_2)\,(1+T_2\,(T_2\,g_{2,(i^+)^+,i}-(-1+T_2)\,g_{2,(j^+)^+,i})-(-1+T_2)\,g_{2,(k^+)^+,i})\,(1+(1-T_1\,T_2)\,g_{3,(k^+)^+,j}+g_{3,(k^+)^+,k}))$$

tex

We do the same for A^r, except this time, without printing at all:

pdf

In[∘]:=
```
Xr = {{1, i, j}, {1, i⁺, k}, {1, j⁺, k⁺}};
Ar = Sum[F₁[c], {c, Xr}] + Sum[F₂[c0, c1], {c0, Xr}, {c1, Xr}];
rhs = Simplify[Ar //. gRules[Xr]];
```

tex

We then compare \verb$lhs$ with \verb$rhs$. The output, \verb$True$, tells us that we have proven~\eqref{eq:R3A}:

pdf

In[∘]:=
```
Simplify[lhs == rhs]
```

Out[∘]=
pdf

```
True
```

tex

We show that $B^l=B^r$ by following exactly the same procedure. Note that we ignore the summation over c_y and instead treat it as fixed. If an equality is proven for every fixed c_y, it is of course also proven for the sum over $c_y\in Y$. Note also that we repeat the test twice, for the two choices of the sign of c_y:

pdf

In[∘]:=
```
Table[
 cy = {s, m, n};
 lhs = Sum[F₂[c0, cy], {c0, Xl}] //. gRules[Xl];
 rhs = Sum[F₂[c0, cy], {c0, Xr}] //. gRules[Xr];
 Simplify[lhs == rhs],
 {s, {1, -1}}
]
```

Out[∘]=
pdf

```
{True, True}
```

tex

Similarly we prove that $C^l=C^r$, and this concludes the proof of Proposition~\ref{prop:R3}.

pdf

In[∘]:=
```
Table[
 cy = {s, m, n};
 lhs = Sum[F2[cy, c1], {c1, Xl}] //. gRules[Xl];
 rhs = Sum[F2[cy, c1], {c1, Xr}] //. gRules[Xr];
 Simplify[lhs == rhs],
 {s, {1, -1}}
]
```

Out[∘]=
pdf

```
{True, True}
```

tex

\ \qed

\begin{remark} \label{rem:E} The computations above were carried out for generic $g_{\nu\alpha\beta}$ and for a generic $c_y=(s,m,n)$; namely, without specifying the knot diagrams in full, and hence without assigning specific values to $g_{\nu\alpha\beta}$, and without specifying m and n. Under these conditions the three parts of~\eqref{eq:ABC} cannot mix (namely, terms from, say, A^h cannot cancel terms in B^h or C^h), and so it would have been enough to show that $E^l=E^r$, where E^h combines A^h and B^h and C^h (and a few harmless further terms) by adding c_y to the summation corresponding to A^h:
\[
 E^h = \sum_{c\in\{c^h_{1,2,3,y}\}} F^h_1(c)
 + \sum_{c_0,c_1\in\{c^h_{1,2,3,y}\}} F^h_2(c_0,c_1).
\]
But that's a simpler computation:

pdf

In[∘]:=
```
ESum[X_] := (Sum[F1[c], {c, X}] + Sum[F2[c0, c1], {c0, X}, {c1, X}]) //. gRules[X];
```

pdf

In[∘]:=
```
Xl = {{1, j, k}, {1, i, k+}, {1, i+, j+}};
Xr = {{1, i, j}, {1, i+, k}, {1, j+, k+}};
Table[
 Simplify[ESum[Append[Xl, {s, m, n}]] == ESum[Append[Xr, {s, m, n}]]],
 {s, {1, -1}}
]
```

Out[∘]=
pdf

```
{True, True}
```

tex

\end{remark}

Proof of R2c

exec

```
nb2tex$TeXFileName = "Invariance-R2c.tex";
```

In[∘]:=
```
X = {{-1, i, j⁺}, {1, i⁺, j}, {1, m, n}};
Simplify[ESum[X]]
```

Out[∘]=

$$\frac{T_1^2 T_2^2 \left(-g_{1,m^+,m} + g_{1,n^+,m} \left(1 - g_{2,m^+,n} + T_2 \left(g_{2,m^+,m} - g_{2,n^+,m}\right) + g_{2,n^+,n}\right)\right) g_{3,n^+,m}}{-1 + T_2} +$$

$$\frac{1}{-1 + T_2} T_1 \Big(T_2^2 \left(g_{1,m^+,m} g_{2,n^+,m} + \left(1 + 2 g_{2,n^+,m} + 2 g_{2,n^+,n}\right) g_{3,m^+,m} - \left(1 - g_{2,m^+,n} + g_{2,n^+,m} + 2 g_{2,n^+,n} + g_{1,n^+,m} \left(1 + g_{2,m^+,m} - g_{2,m^+,n} - g_{2,n^+,m} + g_{2,n^+,n}\right)\right) g_{3,n^+,m}\right) +$$
$$T_2^3 \left(-g_{1,n^+,m} g_{2,m^+,m} g_{3,n^+,m} + g_{2,n^+,m} \left(-g_{3,m^+,m} + \left(1 + g_{1,n^+,m}\right) g_{3,n^+,m}\right)\right) -$$
$$T_2 \left(g_{3,m^+,m} + g_{2,n^+,m} g_{3,m^+,m} + 2 g_{2,n^+,n} g_{3,m^+,m} - g_{2,n^+,n} g_{3,n^+,m} + g_{1,n^+,m} \left(-g_{2,n^+,m} + \left(1 - g_{2,m^+,n} + g_{2,n^+,n}\right) g_{3,n^+,m}\right) + g_{1,m^+,m} \left(g_{2,n^+,m} + g_{2,n^+,n} - g_{3,n^+,m} - g_{3,n^+,n}\right)\right) +$$
$$\left(g_{1,m^+,m} - g_{1,n^+,m}\right) \left(g_{2,n^+,n} - g_{3,n^+,n}\right)\Big) - \frac{g_{3,(j^+)^+,i}}{T_1 T_2} +$$

$$\frac{1}{2 \left(-1 + T_2\right)} \Big(-1 + 2 g_{2,n^+,n} \left(-1 + g_{1,n^+,m} - g_{3,n^+,m}\right) +$$
$$2 T_2^2 \left(-\left(\left(1 + g_{1,n^+,m}\right) g_{2,n^+,m} g_{3,n^+,m}\right) + g_{2,m^+,m} \left(-1 + g_{1,n^+,m} g_{3,n^+,m} - g_{3,n^+,n}\right)\right) -$$
$$2 g_{1,n^+,m} g_{3,n^+,n} + 2 g_{2,n^+,m} g_{3,n^+,n} - 2 g_{3,(j^+)^+,i} +$$
$$T_2 \left(1 + 2 g_{2,n^+,n} + 2 g_{3,n^+,m} - 2 g_{2,m^+,n} g_{3,n^+,m} + 2 g_{2,n^+,m} g_{3,n^+,m} + 4 g_{2,n^+,n} g_{3,n^+,m} - 2 g_{1,n^+,m} \left(g_{2,n^+,m} + \left(-1 + g_{2,m^+,n} - g_{2,n^+,n}\right) g_{3,n^+,m}\right) - 2 g_{2,n^+,m} g_{3,n^+,n} + 2 g_{2,m^+,m} \left(1 + g_{3,n^+,n}\right) + 2 g_{3,(j^+)^+,i}\right)\Big)$$